



DEPARTMENT OF COMPUTING
AND CONTROL ENGINEERING

ДЕПАРТАМЕНТ ЗА РАЧУНАРСТВО
И АУТОМАТИКУ

ACSE

Odsek za automatiku, geomatiku
i upravljanje sistemima

Sub- department for Automatic Control
and System Engineering

Parametarska identifikacija: Određivanje vremenski diskretne funkcije prenosa (deo B)

Modeliranje i simulacija sistema

Generalizovana struktura modela

Generalizovana struktura vremenski diskretnih modela (u Z domenu):

$$A(z)y(t) = \frac{B(z)}{F(z)}u(t) + \frac{C(z)}{D(z)}v(t)$$

$$A(z) = 1 + a_1z^{-1} + \dots + a_nz^{-n}$$


$$B(z) = b_0 + b_1z^{-1} + \dots + b_mz^{-m}$$

$$C(z) = 1 + c_1z^{-1} + \dots + c_rz^{-r}$$


$$D(z) = 1 + d_1z^{-1} + \dots + d_jz^{-j}$$

$$F(z) = 1 + f_1z^{-1} + \dots + f_pz^{-p}$$


Model uključuje merljivi (deterministički) ulaz $u(t)$ i stohastički poremećaj $v(t)$




ARX




ARMAX




OE



BJ



FIR



AR

Nadalje ćemo se posmatrati ARX i ARMAX modele.

ARX - Autoregressive with Extra Input

ARMAX - Autoregressive Moving Average with Extra Input

Ponavljanje: Primer diferencne jednačine 1

Kontinualan linearan model u prostoru stanja

$$\dot{x}(t) = -12.5x(t) + u(t)$$

$$y(t) = 38.9x(t)$$

nakon vremenske diskretizacije periodom odabiranja $T = 0.5$ sekundi postaje

$$x(k + 1) = 0.5353x(k) + 0.03718u(k)$$

$$y(k) = 38.9x(k)$$

odakle je

$$y(k + 1) - 0.5353y(k) = 1.446u(k), \quad k = 0, 1, \dots$$

Ili malo drugačije zapisano

$$y_k = 0.5353y_{k-1} + 1.446u_{k-1}, \quad k = 1, 2, \dots$$

tj.

$$y_k = -a_1y_{k-1} + b_1u_{k-1}, \quad k = 1, 2, 3, \dots, K - 1$$

Ponavljanje: Primer diferencne jednačine 2

Kontinualan linearan model

$$\ddot{y}(t) + 2\dot{y}(t) + 3y(t) = 4u(t)$$

je napisan kao model u prostoru stanja

$$\begin{aligned}\dot{\mathbf{x}}(t) &= \begin{bmatrix} 0 & 1 \\ -3 & -2 \end{bmatrix} \mathbf{x}(t) + \begin{bmatrix} 0 \\ 4 \end{bmatrix} u(t) \\ y(t) &= [1 \quad 0] \mathbf{x}(t)\end{aligned}$$

Nakon vremenske diskretizacije za $T = 0.1$ sek. je dobijen vremenski diskretan model u prostoru stanja

$$\begin{aligned}\mathbf{x}(k+1) &= \begin{bmatrix} 0.986 & 0.09018 \\ -0.2705 & 0.8056 \end{bmatrix} \mathbf{x}(k) + \begin{bmatrix} 0.01868 \\ 0.3607 \end{bmatrix} u(k) \\ y(k) &= [1 \quad 0] \mathbf{x}(k)\end{aligned}$$

od koga se može napraviti funkcija diskretnog prenosa

$$\frac{Y(z)}{U(z)} = W(z) = \frac{0.01868z^{-1} + 0.01748z^{-2}}{1 - 1.792z^{-1} + 0.8187z^{-2}}$$

Odnosno, diferencna jednačina

$$\begin{aligned}(1 - 1.792z^{-1} + 0.8187z^{-2})y(k) &= (0.01868z^{-1} + 0.01748z^{-2})u(k) \\ y_k &= 1.792y_{k-1} - 0.8187y_{k-2} + 0.01868u_{k-1} + 0.01748u_{k-2}\end{aligned}$$

$$y_k = -a_1y_{k-1} - a_2y_{k-2} + b_1u_{k-1} + b_2u_{k-2}, \quad k = 2, 3, \dots, K-1$$

ARX model

- Posmatra se dinamički, linearan proces, čije se ponašanje u okolini stacionarnog stanja može opisati

$$A(z)y(k) = B(z)u(k) + v(k)$$

ili u razvijenom obliku

$$(1 + a_1z^{-1} + \dots + a_nz^{-n})y(k) = (b_0 + b_1z^{-1} + \dots + b_mz^{-m})u(k) + v(k)$$

tada je diferencna jednačina

$$y(k) + a_1y(k-1) + \dots + a_ny(k-n) = b_0u(k) + b_1u(k-1) + \dots + b_mu(k-m) + v(k)$$

gde su inkrementalne vrednosti ulaza $u(k)$ i izlaza $y(k)$, a $v(k)$ je beli šum

$$u(k) = U(k) - U_0, \quad y(k) = Y(k) - Y_0, \quad k = 0, 1, 2, \dots, K-1$$

- Uvođenjem vektora, diferencna jednačina postaje:

$$y(k) = \mathbf{U}^T(k) \cdot \mathbf{q} + v(k), \quad k = 0, 1, 2, \dots, K-1$$


$$\mathbf{U}(k) = [-y(k-1) \quad \dots \quad -y(k-n) \quad u(k) \quad \dots \quad u(k-m)]^T$$

$$\mathbf{q} = [a_1 \quad \dots \quad a_n \quad b_0 \quad \dots \quad b_m]^T$$

- Model je linearan po parametrima i parametri se mogu identifikovati metodom najmanjih kvadrata.

Postupak merenja

- izabere se nominalni radni režim (Y_0, U_0)
- u diskretnim trenucima $t = kT$ se zadaju ulazi i mere izlazi ...
(Napomena: perioda T se bira na osnovu teoreme o odabiranju)
Za svaki diskretan trenutak $k = 0, 1, \dots, K - 1$
 - nominalnom ulazu se doda poznata inkrementalna vrednost $u(k)$,
tj. $U(k) = U_0 + u(k)$ se dovede na ulaz sistema
 - kao ulaz najbolje je upotrebiti pseudoslučajni binarni signal (PRBS - *Pseudo Random Binary Signal*). Vidi sliku: Iskustveno je $u_{max} = 0.05 U_0$.
 - Izmeri se izlaz iz sistema $Y(k)$ i izračuna inkrementalna vrednost izlaza $y(k) = Y(k) - Y_0$



Identifikacija parametara ARX modela

Radi jednostavnosti posmatramo model sa 4 parametra:

$$y(k) = -a_1 y(k-1) - a_2 y(k-2) + b_1 u(k-1) + b_2 u(k-2) + v(k), \quad k = 2, 3, \dots, K-1$$

$$\underbrace{\begin{bmatrix} y(2) \\ y(3) \\ \dots \\ y(K-1) \end{bmatrix}}_{\mathbf{Y}} = \underbrace{\begin{bmatrix} -y(1) & -y(0) & u(1) & u(0) \\ -y(2) & -y(1) & u(2) & u(1) \\ \dots & \dots & \dots & \dots \\ -y(K-2) & -y(K-3) & u(K-2) & u(K-3) \end{bmatrix}}_{\mathbf{S}} \underbrace{\begin{bmatrix} a_1 \\ a_2 \\ b_1 \\ b_2 \end{bmatrix}}_{\mathbf{q}} + \underbrace{\begin{bmatrix} v(2) \\ v(3) \\ \dots \\ v(K-1) \end{bmatrix}}_{\mathbf{N}}$$

Kraće: $\mathbf{Y} = \mathbf{S}\mathbf{q} + \mathbf{N}$, odakle je prema metodi najmanjih kvadrata

$$\hat{\mathbf{q}} = (\mathbf{S}^T \mathbf{S})^{-1} \mathbf{S}^T \mathbf{Y}$$

Realizacija metode za identifikaciju parametara ARX modela

```
function arxident(y, u, n, m)
    # Parametarska identifikacija - ARX model
    # y - vektor merenja izlaza, u - vektor ulaza
    # n - broj koeficijenata A, m - broj koeficijenata B (sa b0)
    # A i B - identifikovani koef. polinoma, J - kriterijum optimalnosti
    # sd - standardna devijacija reziduala

    K = length(u)
    S = zeros(K-n, n+m+1)
    for k = n+1:K
        S[k-n,:] = [-y[k-1:-1:k-n]' u[k:-1:k-m]'];
    end
    Y = y[n+1:K,:];
    q = S \ Y;
    J = 0.5 * (Y-S*q)' * (Y-S*q);
    sd = sqrt(2*J/K);
    A = [1.0; q[1:n]];
    B = q[n+1:n+m+1];
    return A, B, J[1], sd
end
```

pseudoinverzija (S'*S)\S'*Y
kriterijum optimalnosti
kvadratna sredina grešaka
[1 a1 a2 ... an]
[b0 b1 b2 ... bm]

Primer

- ARX model:

$$y(k) - 1.792y(k-1) + 0.8187y(k-2) = 0.01868u(k-1) + 0.01748u(k-2) + v(k)$$

- Generisanje
testnih podataka

```
function generišiMerenja(brojMerenja)
    Random.seed!(1234);
    r = randn(brojMerenja)
    u = (r .> 0) .- 0.5                                # ulaz kao PRBS

    A = [1.0, -1.792, 0.8187]                          # koeficijenti polinoma A i B
    B = [0, 0.01868, 0.01748]
    W = tf(B, A, 1.0)
    y,_,_ = lsim(W, u, 0:length(u)-1)                  # računanje izlaza


    r = randn(brojMerenja);
    sum3 = r * 0.001;    ysum3 = y + sum3;
    sum2 = r * 0.01;     ysum2 = y + sum2;
    sum1 = r * 0.1;      ysum1 = y + sum1;

    return y, u, ysum3, ysum2, ysum1
end


y, u, ysum3, ysum2, ysum1 = generišiMerenja(10000)
```

Primer: izgled prvih 100 podataka...


Ulaz (prvih 100 odbiraka kao PRBS)



Izlaz




Izlaz sa sumom



Primer: identifikovani parametri ARX model-a

```
a, b, J = arxident(y, u, 2, 2)
yst, t, _ = step( tf(b[:],a[:],1.0), 100)
@show("σ=0", a, b, J);
a, b, J = arxident(ysum3, u, 2, 2)
yst3, t, _ = step( tf(b[:],a[:],1.0), 100)
@show("σ=0.001", a, b, J);
a, b, J = arxident(ysum2, u, 2, 2)
yst2, t, _ = step( tf(b[:],a[:],1.0), 100)
@show("σ=0.01", a, b, J);
a, b, J = arxident(ysum1, u, 2, 2)
yst1, t, _ = step( tf(b[:],a[:],1.0), 100)
@show("σ=0.1", a, b, J);
plot(t,[yst yst3 yst2 yst1],
      linetype=:steppost, lab=["original";
                              "sigma=0.001";"sigma=0.01";"sigma=0.1"])
```

Jedinični odzivi dobijenih modela



ARX model	σ šuma	a_1	a_2	b_0	b_1	b_2	J
Bez šuma	0	-1.7920	0.8187	0	0.0187	0.0175	0
Mali šum	0.001	-1.7887	0.8154	0	0.0187	0.0176	0.0232
Srednji šum	0.01	-1.5478	0.5763	-0.0005	0.0189	0.0226	2.0212
Veliki šum	0.1	-0.4897	-0.3649	0.0005	0.0177	0.0462	75.1585

Kašnjenje ulaza ARX modela

- U prethodnom primeru je identifikovan parametar b_0 iako $u(k)$ ne utiče na $y(k)$ – jer kasni.
- Ako se zna da ulaz kasni za $\Delta t = d \cdot T$, onda nije potrebno identifikovati jedan broj b parametara

$$A(z)y(t) = B(z)u(t - \Delta t) + v(t)$$

ili

$$A(z)y(k) = z^{-d}B(z)u(k) + v(k)$$

$$y(k) + a_1y(k-1) + \dots + a_ny(k-n) = b_0u(k-d) + b_1u(k-d-1) \dots + b_mu(k-d-m) + v(k)$$

- Implementacija zahteva da se ubaci dodatni parametar u metodu: kašnjenje d .
- U prethodnom primeru je $d = 1$, što ima za posledicu da se identifikuje tačno 4 parametra umesto ranijih 5.

ARMAX model

- U odnosu na ARX model $A(z)y(k) = B(z)u(k) + v(k)$, gde je $v(k)$ beo šum. ARMAX model je opštiji i pretpostavlja da je šum obojen (beli šum filtriran polinomom $C(z)$).

$$A(z)y(k) = B(z)u(k) + C(z)v(k)$$

$$A(z) = 1 + a_1z^{-1} + \dots + a_nz^{-n}$$

$$B(z) = b_0 + b_1z^{-1} + \dots + b_mz^{-m}$$

$$C(z) = 1 + c_1z^{-1} + \dots + c_rz^{-r}$$

$$n \geq m, \quad n \geq r.$$

- Razvijeni oblik modela:

$$y(k) + a_1y(k-1) + \dots + a_ny(k-n)$$

$$= b_0u(k) + b_1u(k-1) \dots + b_mu(k-m) + v(k) + c_1v(k-1) + \dots + c_rv(k-r)$$

- I ovaj model je linearan po nepoznatim parametrima, ali uz parametre polinoma C ne stoje „poznate“ vrednosti pa se ne može direktno primeniti metoda najmanjih kvadrata.

Identifikacija parametara - Dvokoračni metod

- ARMAX model $A(z)y(k) = B(z)u(k) + C(z)v(k)$ se podeli sa $C(z)$:

$$\frac{A(z)}{C(z)}y(k) = \frac{B(z)}{C(z)}u(k) + v(k)$$

$$\frac{A(z)}{C(z)} \approx H(z) = 1 + h_1z^{-1} + \dots + h_\alpha z^{-\alpha}$$

$$\frac{B(z)}{C(z)} \approx G(z) = g_0 + g_1z^{-1} + \dots + g_\beta z^{-\beta}$$

Teoretski gledano, stepeni α i β su beskonačni, ali je praktično dovoljno:

$$\alpha = 3(n + r)$$

$$\beta = 3(m + r)$$

- **Prvi korak:** na ovaj model se može primeniti metod najmanjih kvadrata

$$H(z)y(k) = G(z)u(k) + v(k)$$

i odrede se parametri polinoma G i H .

- **Drugi korak:** izačuna se greška jednačina

$$e(k) = H(z)y(k) - G(z)u(k), \quad k = 0, 1, \dots, K - 1$$

i upotrebi kao procena za šum $\hat{v}(k) = e(k)$, tako da se iz modela

$$A(z)y(k) = B(z)u(k) + C(z)e(k)$$

metodom najmanjih kvadrata odrede nepoznati parametri polinoma A, B, C

Dvokoračni metod - drugi korak

Posmatra se jednakost:

$$y(k) - e(k) = -a_1 y(k-1) - \dots - a_n y(k-n) \\ + b_0 u(k) + b_1 u(k-1) \dots + b_m u(k-m) + c_1 e(k-1) + \dots + c_r e(k-r).$$

tj. za jedan trenutak k

$$Y(k) = \mathbf{U}^T(k) \mathbf{q}, \quad k = n, n+1, \dots, K-1$$

$$Y(k) = y(k) - e(k) \\ \mathbf{U}^T(k) = [-y(k-1) \quad \dots \quad -y(k-n) \quad u(k) \quad \dots \quad u(k-m) \quad e(k-1) \quad \dots \quad e(k-r)] \\ \mathbf{q} = [a_1 \quad \dots \quad a_n \quad b_0 \quad \dots \quad b_m \quad c_1 \quad \dots \quad c_r]^T$$

Podaci se postave u matrice za sve posmatrane trenutke

$$\mathbf{Y} = \begin{bmatrix} Y(n) \\ Y(n+1) \\ \dots \\ Y(K-1) \end{bmatrix}, \quad \mathbf{S} = \begin{bmatrix} \mathbf{U}^T(n) \\ \mathbf{U}^T(n+1) \\ \dots \\ \mathbf{U}^T(K-1) \end{bmatrix}$$

i

$$\hat{\mathbf{q}} = (\mathbf{S}^T \mathbf{S})^{-1} \mathbf{S}^T \mathbf{Y}$$

```

function armaxident(y,u,n,m,r)
    # Parametarska identifikacija - ARMAX model
    K = length(u)

    # Prvi korak
    alfa = 3*(n+r); beta = 3*(m+r);
    H, G, _ = arxident(y, u, alfa, beta)

    # Drugi korak
    ey,_,_ = lsim( tfz(H, [1.0]), y, 1:length(y))
    eu,_,_ = lsim( tfz(G, [1.0]), u, 1:length(u))
    e = ey .- eu                # procena šuma
    nn = max(r,n);              # za slučaj da je polinom C duži od A
    S = zeros(K-nn, n+m+r+1);
    for i = nn:K-1
        S[i-nn+1,:] = [-y[i:-1:i-n+1]'  u[i+1:-1:i-m+1]'  e[i:-1:i-r+1]']
    end
    Y = y[nn+1:K] - e[nn+1:K]
    q = S \ Y                   # pseudoinverzija
    J = 0.5 * (Y-S*q)' * (Y-S*q) # kriterijum optimalnosti
    sd= sqrt(2*J/K)             # kvadratna sredina grešaka
    A = [1.0; q[1:n]]
    B = q[n+1:n+m+1]
    C = [1.0; q[n+m+2:n+m+r+1]]
    return A, B, C, J[1], sd
end

```

```

function tfz(B, A)
    n = max(length(A), length(B))
    B0 = [B; zeros(Float64,n - length(B))]
    A0 = [A; zeros(Float64,n - length(A))]
    return tf(B0, A0, 1.0)
end

```


Prilagođenje: paket *ControlSystems* posmatra funkcije diskretnog prenosa kao polinome po z , a dok su ovde polinomi po z^{-1} .

Primer: identifikacija ARMAX model-a

```
function testARMAX(r)
    # r je stepen polinoma C
    a, b, c, J = armaxident(y, u, 2, 2, r)
    yst, t, _ = step( tf(b[:],a[:],1.0), 100)
    @show("σ=0", a, b, J)
    a, b, c, J = armaxident(ysum3, u, 2, 2, r)
    yst3, t, _ = step( tf(b[:],a[:],1.0), 100)
    @show("σ=0.001", a, b, J);
    a, b, c, J = armaxident(ysum2, u, 2, 2, r)
    yst2, t, _ = step( tf(b[:],a[:],1.0), 100)
    @show("σ=0.01", a, b, J);
    a, b, c, J = armaxident(ysum1, u, 2, 2, r)
    yst1, t, _ = step( tf(b[:],a[:],1.0), 100)
    @show("σ=0.1", a, b, J);
    plot(t,[yst yst3 yst2 yst1],
        linetype=:steppost,
        lab=["original";"sigma=0.001";
            "sigma=0.01";"sigma=0.1"])
end

testARMAX(2)
testARMAX(6)
```

Jedinični odzivi dobijenih modela



Primer

Rezultati

Dva parametra C polinoma doprinose boljem rezultatu u odnosu na ARX

Veći broj parametara C polinoma i veći broj merenja daju još bolji rezultat

ARX model – uticaj veličine šuma / upotrebljen arxident: $n=2, m=2$

ARX model	σ šuma	a_1	a_2	b_0	b_1	b_2	J
Bez šuma	0	-1.7920	0.8187	0	0.0187	0.0175	0
Mali šum	0.001	-1.7887	0.8154	0	0.0187	0.0176	0.0232
Srednji šum	0.01	-1.5478	0.5763	-0.0005	0.0189	0.0226	2.0212
Veliki šum	0.1	-0.4897	-0.3649	0.0005	0.0177	0.0462	75.1585

ARMAX model – uticaj veličine šuma / upotrebljenl armaxident: $n=2, m=2, r=2$

ARMAX model	σ šuma	a_1	a_2	b_0	b_1	b_2	J
Bez šuma	0	-1.7920	0.8187	0	0.0187	0.0175	0
Mali šum	0.001	-1.7916	0.8183	0	0.0187	0.0175	0.0002
Srednji šum	0.01	-1.7857	0.8125	0	0.0186	0.0181	0.0186
Veliki šum	0.1	-1.6092	0.6395	0	0.0177	0.0254	1.2631

ARMAX model – uticaj broja merenja / upotrebljen armaxident: $n=2, m=2, r=6$

Broj odbiraka	σ šuma	a_1	a_2	b_0	b_1	b_2	J	σ reziduala
10^4	0.1	-1.6821	0.7101	0.0005	0.0173	0.0242	0.5634	0.0106
10^5	0.1	-1.7760	0.8027	-0.0005	0.0202	0.0171	0.9235	0.0043
10^6	0.1	-1.7836	0.8105	-0.0001	0.0187	0.0179	5.3500	0.0033

Kašnjenje ulaza ARMAX modela

- Kao što je objašnjeno kod ARX modela, poznavanje kašnjenja ulaza smanjuje broj nepotrebnih parametara koji se identifikuju u polinomu $B(z)$ (jer su nula)

$$A(z)y(k) = B(z)u(t - \Delta t) + C(z)v(k)$$

ili

$$A(z)y(k) = z^{-d}B(z)u(k) + C(z)v(k)$$

- Pogledati implementaciju.

Određivanje reda AR(MA)X modela

Generalno, sa povećavanjem broj parametara $\mathcal{J}$ se smanjuje, ali kada je broj identifikovanih parametara veći od potrebnog smanjenje $\mathcal{J}$ je malo. Ovo se može upotrebiti za procenu reda polinoma A, B .

Vrednosti $\mathcal{J}$ je prikazana u tabelama. Šum „srednji“ $\sigma = 0.01$

Nema jasnog pada $\mathcal{J}$. →

ARX model		n					
		1	2	3	4	5	6
m	1	504.0	267.9	236.7	236.3	236.0	235.9
	2	-($m > n$)	210.5	149.9	146.5	146.3	146.1
	3	-	-	130.6	108.8	108.1	107.0
	4	-	-	-	100.0	89.4	89.2

Pad $\mathcal{J}$, ukazuje da je verovatno dobar izbor: $n = 2, m = 2$ (u oba slučaja)

ARMAX model $r=3$		n					
		1	2	3	4	5	6
m	1	406.4	36.55	20.15	17.15	16.95	16.91
	2	-	1.0833	0.9203	0.9127	0.9122	0.9115
	3	-	-	0.4215	0.3505	0.3508	0.3508
	4	-	-	-	0.1733	0.1397	0.1392

ARMAX model $r=6$		n					
		1	2	3	4	5	6
m	1	405.5	34.84	17.39	11.29	8.22	6.37
	2	-	0.0676	0.0539	0.0570	0.585	0.582
	3	-	-	0.0102	0.0082	0.0092	0.0091
	4	-	-	-	0.0084	0.0094	0.0092

Rekurzivna metoda najmanjih kvadrata (RMNK) - ideja

- Posmatramo k -ti trenutak (nakon k merenja) i tada su procenjeni parametri sistema

$$\hat{\mathbf{q}}(k) = (\mathbf{S}^T(k)\mathbf{S}(k))^{-1}\mathbf{S}^T(k)\mathbf{Y}(k)$$

gde je

$$\mathbf{Y}(k) = \begin{bmatrix} y(1) \\ y(2) \\ \vdots \\ y(k) \end{bmatrix}, \quad \mathbf{S}(k) = \begin{bmatrix} \boldsymbol{\phi}^T(1) \\ \boldsymbol{\phi}^T(2) \\ \vdots \\ \boldsymbol{\phi}^T(k) \end{bmatrix}$$

- Ako se uzme u obzir i dodatno merenje iz narednog trenutka

$$\hat{\mathbf{q}}(k+1) = (\mathbf{S}^T(k+1)\mathbf{S}(k+1))^{-1}\mathbf{S}^T(k+1)\mathbf{Y}(k+1)$$

gde je

$$\mathbf{Y}(k+1) = \begin{bmatrix} \mathbf{Y}(k) \\ y(k+1) \end{bmatrix}, \quad \mathbf{S}(k+1) = \begin{bmatrix} \mathbf{S}(k) \\ \boldsymbol{\phi}^T(k+1) \end{bmatrix}$$

- Ideja: Za određivanje $\hat{\mathbf{q}}(k+1)$ upotrebiti postojeću procenu parametara $\hat{\mathbf{q}}(k)$ i korigovati je na osnovu novih merenja izlaza $y(k+1)$ i ulaza (dodanog u $\boldsymbol{\phi}^T(k+1)$).

Rekurzivna metoda najmanjih kvadrata

- Nova procena parametara $\hat{\mathbf{q}}(k + 1)$ se dobija na osnovu procene prethodne procene $\hat{\mathbf{q}}(k)$

$$\underbrace{\hat{\mathbf{q}}(k + 1)}_{\substack{\text{tekuća} \\ \text{procena} \\ q}} = \underbrace{\hat{\mathbf{q}}(k)}_{\substack{\text{prethod.} \\ \text{procena} \\ q}} + \underbrace{\mathbf{P}(k + 1)\boldsymbol{\phi}(k + 1)}_{\text{korekcionni vektor}} \left(\underbrace{y(k + 1)}_{\substack{\text{novo} \\ \text{merenje}}} - \underbrace{\boldsymbol{\phi}^T(k + 1)\hat{\mathbf{q}}(k)}_{\text{predviđeno merenje}} \right)$$

gde se matrica $\mathbf{P}$ takođe računa rekurzivno:

$$\mathbf{P}(k + 1) = \mathbf{P}(k) - \mathbf{P}(k)\boldsymbol{\phi}(k + 1) \cdot \left(\mathbf{I} + \boldsymbol{\phi}^T(k + 1)\mathbf{P}(k)\boldsymbol{\phi}(k + 1) \right)^{-1} \boldsymbol{\phi}^T(k + 1)\mathbf{P}(k)$$

- Početne vrednosti su: $\hat{\mathbf{q}}(0) = \mathbf{0}$ i $\mathbf{P}(0) = c\mathbf{I}$ (gde je c veliki broj)
- Koraci:
 1. prikupe se nove vrednosti ulaza $u(k + 1)$ i izlaza $y(k + 1)$
 2. formiraju se $\boldsymbol{\phi}(k + 1)$
 3. izračuna se $\mathbf{P}(k + 1)$
 4. izračuna se procena parametara $\hat{\mathbf{q}}(k + 1)$
 5. memorišu se: $\mathbf{P}(k) = \mathbf{P}(k + 1)$ i $\hat{\mathbf{q}}(k) = \hat{\mathbf{q}}(k + 1)$ za naredni trenutak odabiranja

Identifikacija promenljivih parametara modela

- Ako je izlaz modela linearan po nepoznatim parametrima, onda je rekurzivna metoda najmanjih kvadrata pogodna za *on-line* parametarsku identifikaciju
- Ako se parametri modela menjaju tokom rada sistema onda je potrebno vršiti parametarsku identifikaciju u realnom vremenu.
 - Tada je pogodno dati veću važnost skorijim merenjima, a davnašnja merenja potisnuti.
Kriterijum optimalnosti koji ovo uzima u obzir:

$$\mathcal{J}_\rho = \frac{1}{2} \sum_{i=0}^k \rho^{k-i} (y(i) - \mathbf{U}^T(i)\mathbf{q})^2, \quad 0 < \rho \leq 1$$

gde je ρ faktor zaboravljanja (tipično je $0.98 < \rho \leq 1$)

- manja vrednost ρ omogućava „brže zaboravljanje“ starih merenja.
- Korekcija $\mathbf{P}$ kada postoji faktor zaboravljanja
$$\mathbf{P}(k+1) = \frac{1}{\rho} \left(\mathbf{P}(k) - \frac{\mathbf{P}(k)\boldsymbol{\phi}(k+1)\boldsymbol{\phi}^T(k+1)\mathbf{P}(k)}{\rho + \boldsymbol{\phi}^T(k+1)\mathbf{P}(k)\boldsymbol{\phi}(k+1)} \right)$$

Realizacija RMNK

```
using LinearAlgebra

function arxidentrek(y, u, n, m)
    # Parametarska identifikacija - rekurzivni metod najmanjih kvadrata
    # NAPOMENA: algoritam radi sa ranije snimljenim y i u
    # (nije on-line realizacija)
    # može se dodati dodatni ulazni parametar za kašnjenje ulaza

    ρ = 0.995
    q = zeros(n+m+1)
    P = 1000.0 * Diagonal(ones(n+m+1))
    tacaka = length(u);
    for k = n+1:tacaka
        U = [-y[k-1:-1:k-n]; u[k:-1:k-m]]
        P = (P - P*U*U'*P / (ρ + U'*P*U)) / ρ
        q += P*U*(y[k]-U'*q)
    end
    A = [1.0; q[1:n]]
    B = q[n+1:n+m+1]
    return A, B
end
```


Primer: RMNK - podaci

- Simulacija: posle 500 sekundi (5000 odbiraka) model W1 se promenio u model W2
 - W1: $y(k+1) - 0.5353y(k) = 1.446u(k) + v(k)$
 - W2: $y(k+1) - 0.4607y(k) = 1.353u(k) + v(k)$

```
T = 0.1
W1 = ss(0.5353, 1.446, 1, 0, T)
W2 = ss(0.4607, 1.353, 1, 0, T)

K = 5000
r = randn(K)
u1 = (r .> 0) .- 0.5
t = 0:T:(K-1)*T
y1, _, x1 = lsim(W1,u1,t)
x0 = x1[end]
r = randn(K)
u2 = (r .> 0) .- 0.5
y2, _ = lsim(W2,u2,t;x0=[x0])
šum = randn(2*K) * 0.1
u = [u1; u2]
y = [y1; y2]
yšum = y + šum;
```

Primer: RMNK - identifikacija

- Referentni modeli (dati radi poređenja dole prikazanih identifikovanih parametara):
 - W1: $y(k + 1) - 0.5353y(k) = 1.446u(k) + v(k)$
 - W2: $y(k + 1) - 0.4607y(k) = 1.353u(k) + v(k)$

```
julia> a, b = arxidentrek(yšum[1:K],u[1:K],1,1)
([1.0, -0.5305383362090255], [0.007686638174982491, 1.4254728432453667])

julia> a, b = arxidentrek(yšum,u,1,1)
([1.0, -0.45281229151779456], [0.007064469842641146, 1.336883365611286])
```

Identifikovanje parametara
upotrebom prve polovine
merenja (dok se model nije
promenio).
Dobro poklapanje sa W1.

Upotrebljena su sva
merenja, gde je uključena
promena modela.
Dobro poklapanje sa W2
(zbog faktora zaboravljanja).

Real-time realizacija RMNK

Funkcija **IdentRek** je namenjena za izvršavanje u realnom vremenu. Najvažnija ugnježdjena metoda te funkcije je **ident**. Ona se poziva svake periode odabiranja da bi se ostvarila identifikacija u realnom vremenu.

```
function IdentRek(n_, m_, d_)
    n::Int = n_           # broj A parametara ARX modela
    m::Int = m_           # broj B parametara ARX modela
    d::Int = d_           # kasnjenje
    # inicijalizacija
    q = zeros(Float64, n+m) # parametri
    P = Diagonal(ones(n+m)*1000.0) # pomoćna matrica
    UY = zeros(n)          # poslednju izlazi
    UU = zeros(m+d)        # poslednji ulazi

    getA() = [1; q[1:n]]
    getB() = [zeros(Float64,d); q[n+1:n+m]]

    function ident(y, u)
        UU = [u; UU[1:m+d-1]] # ubaci novi ulaz
        U = [UY; UU[d+1:d+m]] # poveži izlaze i ulaze
        ρ = 0.995               # mogao bi biti parametar metode
        P = (P - P*U*U'*P / (ρ + U'*P*U)) / ρ
        q += P*U*(y-U'*q)
        UY = [-y; UY[1:n-1]]   # ubaci novi izlaz
        return q
    end

    () -> (getA, getB, ident)
end
```

Primer: Real-time realizacija RMNK

```
julia> id = IdentRek(1,1,1);      # inicijalizacija
julia> Q = [0.0 0.0];           # početna procena parametara
julia> for i=1:10000             # po svim trenucima
    q = id.ident(y[i],u[i])      # identifikacija u RT
                                # ili ...(yšum[i]...
    global Q = [Q; q']          # memorisanje parametara
end;

julia> A = id.getA();           # očitavanje identifikovanih parametara
julia> B = id.getB();
julia> @show(A,B);
A = [1.0, -0.460700000000111594]
B = [0.0, 1.35300000000010738]

julia> plot(1:10001, Q, lab=["a_1","b_1"],
           title="Promene parametara bez suma")
```

Primer upotrebe **IdentRek** na podacima gde nema šuma.

Primer upotrebe **IdentRek** na podacima gde IMA šuma.

